

# Differentiaalvergelijkingen met Maple

Handleiding

---

Oktober 1995      Originele versie : ir P.J. Dekker  
September 2000    Aangepast door : dr ir W.T. van Horssen & ir C.W.J. Lemmens  
Februari 2001     Aangepast door : ir C.W.J. Lemmens

## Voorwoord

Een **C**omputer **A**lgebra **S**ysteem (CAS) is een krachtig instrument waarmee een (toekomstig) ingenieur een veelheid aan problemen kan oplossen. Grofweg kan een CAS op drie manieren gebruikt worden:

- voor het oplossen van eenvoudige problemen
- als hulpmiddel bij meer complexe problemen
- om oplossingen te visualiseren

Deze handleiding leert u een CAS gebruiken bij het oplossen van gewone en partiële differentiaalvergelijkingen. Voor een grote klasse van gewone differentiaalvergelijkingen kan op een eenvoudige wijze een oplossing worden verkregen. Het zal echter blijken dat enige voorzichtigheid geboden is: de antwoorden zijn niet altijd volledig of correct.

Bij het oplossen van partiële differentiaalvergelijkingen moet de gebruiker het probleem eerst in deelproblemen opsplitsen. Het CAS kan daarna gebruikt worden bij de verschillende stappen, en tot slot om de oplossing, vaak bestaand uit ingewikkelde integralen en sommaties, in beeld te brengen.

Deze handleiding laat u kennis maken met de (on)mogelijkheden van het CAS Maple op het gebied van de differentiaalvergelijkingen. Een korte introductie in het interactieve gebruik van Maple maakt deel uit van de handleiding.

De handleiding is op dit moment gebaseerd op Maple versie V Release 5.1 voor Linux. De Technische Universiteit Delft is sinds enige jaren in het bezit van een campuslicentie voor Maple, waardoor dit pakket voor iedere student of medewerker van de TU voor een gering bedrag bij Campuslicentiemarkt te verkrijgen is (zowel voor Unix/Linux als Windows 98/NT).

(Meer informatie over de campuslicentie: <http://www.dto.tudelft.nl/homecls.htm>)

Bij verschillende practica wordt gebruik gemaakt van voorbeeld-files, de zogeheten *Maple-worksheets*. Deze zijn meestal ook via het WWW op te halen. De exacte locatie zal dan tijdens het praktikum door de docent worden medegedeeld.

Veel succes!

# Inhoudsopgave

<b>1</b>	<b>Algemeen</b>	<b>3</b>
1.1	Wat is Maple? . . . . .	3
1.2	Werken met Maple . . . . .	3
1.3	Worksheets . . . . .	4
<b>2</b>	<b>Leer Maple in 60 minuten!</b>	<b>5</b>
2.1	Het samenstellen van een Maple-commando . . . . .	5
2.2	Oplossen van vergelijkingen . . . . .	6
2.3	Manipuleren en vereenvoudigen . . . . .	7
2.4	Integreren . . . . .	8
2.5	Differentiëren . . . . .	9
2.6	Plotten . . . . .	9
2.7	Functies . . . . .	10
<b>3</b>	<b>Gewone differentiaalvergelijkingen</b>	<b>12</b>
3.1	<code>dsolve</code> : de differential equation solver . . . . .	12
3.2	Plotten . . . . .	13
3.3	Picard iteratie . . . . .	14
3.4	De Laplace transformatie . . . . .	14
3.5	Stelsels differentiaalvergelijkingen . . . . .	15
<b>4</b>	<b>Partiële differentiaalvergelijkingen</b>	<b>18</b>
4.1	Algemene commando's . . . . .	18
4.2	Transformatie van variabelen . . . . .	18
4.3	Fourier reeksen . . . . .	18
4.4	Eerste orde partiële differentiaalvergelijkingen . . . . .	19
	<b>Alle gebruikte commando's op een rijtje</b>	<b>21</b>

# 1 Algemeen

## 1.1 Wat is Maple?

Maple is, zoals reeds vermeld, een computer algebra systeem. Dergelijke systemen onderscheiden zich van andere wiskundige pakketten door de mogelijkheid om exact te rekenen. Zo is  $\cos \frac{\pi}{6}$  gelijk aan  $\frac{1}{2}\sqrt{3}$  en niet 0.8660254. Is exact rekenen niet (langer) mogelijk of wenselijk, dan kan eenvoudig overgegaan worden op numerieke benaderingen.

Daarnaast kan er ook met formules gerekend worden:  $\int \cos x dx$  wordt  $\sin x$ , en  $(a + b)(a - b)$  kan uitgewerkt worden tot  $a^2 - b^2$ .


Er zijn twee soorten computer algebra systemen: systemen speciaal ontwikkeld voor gebruik op een deelgebied van de wiskunde en systemen voor algemeen gebruik. Maple is een systeem voor algemeen gebruik op uiteenlopende gebieden van de wiskunde zoals analyse, lineaire algebra, statistiek, lineaire optimalisering, groepentheorie, enz. . Andere bekende algemene systemen zijn Mathematica, Reduce, Derive en Axiom.

Sinds 1980 wordt Maple ontwikkeld aan de Universiteit van Waterloo in Ontario, Canada. Maple is de naam van een boom, de esdoorn. Het blad van deze boom komt dan ook terug in het logo van Maple. Esdoornen komen veel voor in Canada, en het esdoornblad is daardoor het nationale symbool van Canada. De vlag van Canada bevat ook een Maple-blad.

## 1.2 Werken met Maple

Het pakket Maple draait zowel onder Linux als Windows, maar voor het praktikum wordt Linux gebruikt omdat dit betrouwbaarder werkt en eenvoudiger te beheren is. Informatie over het werken met Linux is o.a. beschikbaar bij de praktikumorganisatie.

Na het inloggen onder Linux verschijnt ongeveer het volgende window :



Achter de `>`, de Maple-prompt, kunt u commando's intoetsen. Deze worden dan direct door Map-

le uitgevoerd. Het window waarin Maple draait kunt u met de muis of de middelste knop in de rechterbovenhoek van het scherm kleiner of groter maken.

### 1.3 Worksheets

Aan het begin van een sessie moet het in de vorige paragraaf beschreven window verschijnen. Vervolgens kunt u Maple-commando's gaan invoeren. U kunt uw Maple-sessie, een zogeheten *worksheet*, bewaren in uw eigen homedirectory op het systeem. Kies in het menu [**F**ile] de optie [**S**ave **A**s] en voer een geschikte filenaam in. Let erop dat u in een directory staat waar u ook schrijfrechten heeft !

Bij de volgende sessie kunt u dan het worksheet met [**F**ile], [**O**pen] weer ophalen.

Voor sommige praktica zijn er voorbeeldfiles aanwezig. Deze kunt u op dezelfde wijze openen, waarbij echter vaak eerst naar een aparte subdirectory moet worden geklikt.

Het is ook mogelijk relevante plotjes of worksheets op de printers af te drukken. Meer informatie over het gebruik van de printers is bij de praktikumorganisatie te verkrijgen.

Met de pull-down menu's bovenaan de window kunnen alle mogelijke bewerkingen op een worksheet gedaan worden. Belangrijke toetsen zijn de “ $\Sigma$ ” , “**T**” en “[>” toetsen : daarmee kan worden omgeschakeld naar resp. Maple mathmode, Maple tekstmode en Maple math-invoermode.

Mocht de uitvoer van een commando te lang op zich laten wachten, dan kunt u met de toets met het **STOP** symbool de bewerkingen afbreken en de prompt weer terug krijgen.

Variabelen en plotjes kosten geheugenruimte. Het is daarom aan te raden om plot-windows af te sluiten en niet naar de achtergrond te schuiven of te klikken. Afsluiten gaat het eenvoudigst door tweemaal met de linker muis-toets op het streepje linksboven in het window te klikken.

M.b.v. de menu-optie **Window**  $\rightarrow$  **Cascade** kunnen *alle* windows met “tabbladen” zichtbaar gemaakt worden, waarna een window naar voren gehaald kan worden door in de bovenbak van dat window te klikken.

Het geheugen kan schoongemaakt worden met het **restart:** commando. Alle tekst blijft dan op het scherm maar alle variabelen worden geïnitieerd. Door de benodigde commando's nogmaals uit te voeren kunt u uw berekeningen voortzetten. Het is dan voldoende om de cursor met hulp van de muis ergens op de betreffende regel te zetten en daarna op [**R**eturn] of [**E**nter] te drukken.

Een apart window binnen Maple sluiten gebeurt NIET met een klik op het meest **rechtste** knopje in de bovenbalk, maar juist door het meest **linkse** knopje aan te klikken en dan “close” te kiezen. Dit kan overigens ook m.b.v. de menu-optie **File**  $\rightarrow$  **Close**.

## 2 Leer Maple in 60 minuten!

In dit hoofdstuk leiden we u langs de belangrijkste Maple-commando's. U zult merken dat het werken met Maple enig geduld en nauwkeurigheid vereist, maar in het algemeen hebben de commando's een heldere en logische opbouw. Na 60 minuten kunt u zelf eenvoudige (differentiaal-)vergelijkingen oplossen en enkele mooie plaatjes maken. Mocht u onverhoopt de werking van een bepaald commando vergeten zijn dan kunt u te allen tijde met

> ?commando

of met behulp van de *help-browser* de betreffende help-file opvragen.

### 2.1 Het samenstellen van een Maple-commando

Het intikken van opdrachten is een nauwkeurig werkje. Als u een \* of een ) vergeet, begrijpt Maple de opdracht verkeerd of in het geheel niet. We geven hier een lijst met links de in de wiskunde gebruikelijke notaties en rechts de vorm waarin het ingetikt moet worden. Op het scherm verschijnt het resultaat doorgaans in de wiskundige notatie.

Wiskundige notatie	Intikken
$a + b$	a+b
$a - b$	a-b
$ab$	a*b
$3xy$	3*x*y
$\frac{a}{b}$	a/b
$a^b$	a^b
$\sqrt{a}$	sqrt(a)
$ x $	abs(x)
$\pi$	Pi
$e$	E
$i$	I
$a + bi$	a+b*I
$\infty$	infinity
$\sin x, \tan x, \dots$	sin(x), tan(x), ...
$e^x$	exp(x)

#### Opmerkingen:

- hoeken moeten altijd in radialen genoteerd worden.
- pi wordt net als Pi door  $\pi$  weergegeven, echter: pi is een variabele, Pi de konstante (3.1415...).
- ook andere griekse letters kunnen in Maple gebruikt worden. Zo wordt de letter lambda in uitvoer als  $\lambda$  weergegeven. Enige voorzichtigheid is geboden omdat sommige letters voor Maple gereserveerd zijn, te weten:  
Beta, gamma, GAMMA, Chi, Psi, Pi, Zeta.

Maple commando's worden voor de duidelijkheid voorafgegaan door >, de prompt. Dit teken staat al op uw scherm en moet dus niet worden ingetikt. De commando's staan ook in een ander lettertype, bijvoorbeeld: `dsolve({dv,y(0)=0},y(x))`

Opdrachten in Maple moeten worden afgesloten door een ; (puntkomma, semicolon) of een : (dubbele punt, colon). Als u na een ; de [Return]- of [Enter]-toets indrukt, wordt de opdracht ingevoerd en de reactie van Maple verschijnt op het scherm. Als u na een : de [Return]- of [Enter]-toets indrukt, wordt de opdracht ingevoerd en door Maple verwerkt. Er verschijnt echter geen reactie op het scherm.

Wanneer u een puntkomma vergeet dan kunt u deze alsnog op een volgende regel invoeren. Gaat u echter met de cursor terug naar de betreffende regel en plaatst u daar alsnog een puntkomma dan geeft Maple de foutmelding **syntax error**. Ga nogmaals terug en tik op **[Return]** of **[Enter]**: het commando wordt dan correct uitgevoerd.

Laten we beginnen: wat verwacht u als antwoord op de volgende commando's?

```
> 36/6;
> 36/8;
> 36^(1/4);
> simplify(%);
> 4^(4^4);
> evalf(%);
```

Maple rekent dus echt exact, eventueel tot op 500000 cijfers nauwkeurig! Soms wordt het antwoord direct vereenvoudigd, soms moet u helpen met het commando **simplify**. Om een benadering te krijgen gebruiken we **evalf**, **evaluate** in floating point.

De procenttekens % betekenen dat hier het resultaat van de voorgaande opdracht ingevuld moet worden; dit bespaart veel tikwerk. Op dezelfde wijze kunt u met %% en %%% de resultaten van de opdracht daarvoor en die daar weer voor invoeren. Verder teruggaan is niet mogelijk. Let op: % is het resultaat van de laatst uitgevoerde opdracht, waar die ook op uw scherm staat!

Let bij het invoeren van gecompliceerde getallen en formules vooral op het plaatsen van haakjes:

```
> 8^-2/3;
> 8^(-2/3);
> 8^(-2)/3;
```

leveren alle een ander antwoord.

Met := kunt u getallen, vergelijkingen of uitkomsten aan variabelen toekennen:

```
> f := x^2+4*sin(x);
```

Als u nu een formule met een **f** intikt, dan vervangt Maple deze automatisch door  $x^2 + 4\sin(x)$ :

```
> arctan(f);
```

Al deze variabelen worden in het geheugen bewaard tot het einde van een sessie. Op de volgende manier kunt u **f** een andere waarde geven:

```
nieuwe toekenning > f := x^3;
unassign           > f := 'f';
```

U kunt geen toewijzingen doen aan interne variabelen als **I** of **E**.

Door simpelweg de naam van de variabele, gevolgd door ; in te tikken kan zijn waarde opgevraagd worden.

## 2.2 Oplossen van vergelijkingen

Vergelijkingen zijn zelden exact op te lossen. Meestal moet uitgeweken worden naar methoden waarmee de nulpunten numeriek benaderd kunnen worden.

Zo kan Maple de polynoomvergelijking

```
> f := x^2 + 5*x - 3 = 0;
> solve(f,x);
```

wel oplossen, maar de vergelijking

```
> f := sin(x/2-1) = x/10;  
> solve(f,x);
```

niet. Met het commando `fsolve` (inderdaad, floating point `solve` ) kunnen we nulpunten numeriek benaderen,

```
> fsolve(f,x);
```

maar compleet is dit antwoord niet. In veel gevallen zult u zelf een handje moeten helpen door bijvoorbeeld een interval te geven waarop naar de oplossing gezocht moet worden.

```
> fsolve(f,x,-5..0);  
> fsolve(f,x,5..10);
```

Het antwoord op

```
> f := x^3-5*x^2+x-1 = 0;  
> solve(f,x);
```

zal u waarschijnlijk enkele hoofdbreken kosten. Om het antwoord in te korten heeft Maple `% 1` gesubstitueerd, waarbij `% 1 :=  $\frac{116}{27} + \frac{2}{9}\sqrt{26}\sqrt{3}$`  is. Benader deze waarde met `evalf` en los dan de vergelijking eens op met `fsolve`.

Maple kan ook stelsels oplossen. De vergelijkingen moeten dan tussen accolades geplaatst worden en de gevraagde onbekenden ook. Als voorbeeld bepalen we de snijpunten van de eenheidscirkel  $x^2 + y^2 = 1$  met de ellips  $\frac{x^2}{4} + 2y^2 = 1$  (in dit geval zijn dat er 4) :

```
> stelsel := {x^2 + y^2 = 1, x^2/4 + 2*y^2 = 1};  
> solution := solve(stelsel,{x,y});
```

Maple geeft uitdrukkingen m.b.v. “RootOf” waardoor een compacte notatie wordt gewaarborgd. Wil je alle mogelijke waarden weten, gebruik dan het “allvalues” commando.

Met “implicitplot” (zie sectie 3.2 op pagina 13) kun je van het probleem ook een plaatje maken (Let op: gewoon “plot” werkt niet omdat er hier immers geen sprake is van functies !)

```
> allvalues(solution,{x,y});      # (Oude Maple versies)  
> allvalues(solution);           # (Maple vanaf versie V release 5.1)  
> implicitplot(stelsel,x=-2..2,y=-2..2);
```

### 2.3 Manipuleren en vereenvoudigen

Zoals in de inleiding al is vermeld is de interessantste mogelijkheid van computer-algebra programma’s dat ze niet alleen met getallen kunnen rekenen, maar ook met formules. We geven hier ter demonstratie een paar voorbeelden van Maple commando’s voor formulemanipulatie.

Het commando `expand` (ontwikkel) toegepast op een polynoom werkt de haakjes weg. Voer in

$$(x^2 + 1)(2x - 1)(x^3 + 8)$$

en geef daarna de opdracht

```
> expand(%);
```

en Maple heeft de haakjes weggewerkt. De volgorde van de termen is niet de gebruikelijke, maar na:



```
> sort(%);
```

staan de termen gerangschikt naar afnemende machten van  $x$  zoals u gewend bent.  
De tegenhanger van `expand` is `factor`. Tik in:

```
> factor(%);
```

en het polynoom wordt weer ontbonden. U ziet dat nu ook  $x^3 + 8$  ontbonden is. Maple ontbindt hier niet verder dan in polynomen met gehele coëfficiënten.

Voor het vereenvoudigen van goniometrische formules zijn er meerdere mogelijkheden. Met

```
> combine(cos(x)^2, trig);
```

gaan we over op de dubbele hoek, en met `expand` weer terug.

```
> simplify(sin(x)^2+cos(x)^2);
```

doet de voor de hand liggende vereenvoudiging.

Door een substitutie uit te voeren kan een probleem ook vereenvoudigd worden.

```
> f := (x+y)^2 + 1/(x+y)^2;
```

```
> subs(x+y=z, f);
```

Met `normal` worden de breuken onder één noemer gebracht.

```
> normal(%);
```

```
> subs(z=x+y, f);
```

Onderzoek ook de werking van de commando's `simplify` en `expand` op dit probleem eens.

## 2.4 Integreren

Alhoewel nog niet alles feilloos werkt, is integratie toch één van de succesnummers van de computer-algebra.

Integreren gebeurt door het commando `int` met vermelding van de variabele waarnaar geïntegreerd moet worden.

```
> Int(cos(a*x), x) = int(cos(a*x), x);
```

Op het scherm verschijnt als antwoord:

$$\int \cos(ax) dx = \frac{\sin(ax)}{a}$$

Dit is inderdaad een primitieve van  $\cos(ax)$ . Zoals u ziet laat Maple de constante weg (en maakt er ook geen punt van dat  $a \neq 0$  moet zijn.)

Het zal duidelijk zijn dat het commando `Int` de ongeëvalueerde integraal geeft. Door het `=`-teken kunnen we het probleem mooi op het scherm brengen. Waarom gebruiken we hier `:=` niet?

Zoals u weet zijn niet alle onbepaalde integralen uit te drukken in elementaire functies. Op

```
> int(sin(x^3), x);
```

weet ook Maple geen antwoord.

Een bepaalde integraal wordt uitgerekend door in de opdracht de grenzen van de integraal te vermelden

```
> int(sin(x), x=0..Pi);
```

Maple heeft geen problemen met oneigenlijke integralen. Rekent u eens uit

$$\int_2^{\infty} \frac{1}{x \ln^2 x} dx$$

( $\infty$  wordt aangeduid met `infinity`).

## 2.5 Differentiëren

Differentiëren gebeurt door het commando `diff` met vermelding van de variabele waarnaar gedifferentieerd moet worden.

```
> f:=x^y;  
> diff(f,x);
```

Ook mogelijk is

```
> diff(f,y);
```

of

```
> diff(f,x,y);
```

Probeer eens:

```
> diff(f,y,x);
```

De tweede en hogere afgeleiden van  $f$  krijgt u met het commando `diff(f,x$2)`, `diff(f,x$3,y$2)`, etc.

## 2.6 Plotten

Stel we willen de functie

$$\sqrt[3]{x^2 - x^3}$$

tekenen. We voeren dan eerst de functie in door:

```
> f := (x^2-x^3)^(1/3);
```

Daarna geven we het commando:

```
> plot(f,x=-3..3,-2..2);
```

en op het scherm verschijnt de grafiek van  $f$  voor  $-3 \leq x \leq 3$ , waarbij de  $y$ -as loopt van  $-2$  tot  $2$ .

U moet aangeven voor welke  $x$  u de grafiek wilt zien; hier is gekozen voor  $x = -3..3$ . U kunt het bereik, hier  $-2..2$ , weglaten in de opdracht. Maple kiest dan zelf een bereik. Dat gaat echter mis als de functie een verticale asymptoot heeft.

```
> h := 1/x;  
> plot(h,x=-5..5);
```

Boven het plot-window staan verschillende menu's waarmee u onder andere het assen-stelsel en het lijn-type kunt instellen. Probeer de verschillende mogelijkheden eens uit.

U kunt ook 2 (of meer) functies tegelijk tekenen, bijvoorbeeld om ze te vergelijken. Voer in:

```
> f := sin(x);  
> g := x - x^3/3!;
```

en geef daarna de opdracht:

```
> plot({f,g},x=-Pi..Pi);
```

Het paar  $f, g$  moet tussen accolades geplaatst worden. Op het scherm verschijnen de grafiek van  $f$  en  $g$  in verschillende kleuren. Er staat echter niet bij welke de grafiek van  $f$  en welke die van  $g$  is.

In drie dimensies is het aantal mogelijkheden nog veel groter. Probeer aan de hand van de volgende functie de verschillende menu's eens uit. Na een klik met de rechter muis-toets wordt de grafiek met de nieuwe opties getekend.

```
> f := exp(-x^2-y^2)*sin(Pi*x)*sin(Pi*y);
> plot3d(f, x=-2..2, y=-2..2);
```

Als optie kunt u het aantal punten opgeven dat in  $x$ - en  $y$ -richting getekend moet worden:

`grid=[40,40]`. Dit levert een gedetailleerdere grafiek, maar kost wel meer rekentijd!

Het oppervlak kan van alle kanten bekeken worden. Klik met de linker muis-toets ergens in het window: er verschijnt dan een kubus in beeld. Houd nu de linker muis-toets ingedrukt en beweeg door het window. Na het indrukken van de rechter muis-toets wordt de grafiek opnieuw getekend.

Zowel 2D als 3D plotjes kunnen achter elkaar geplaatst worden om een animatie-filmpje te vormen. Het commando

```
> animate(sin(a*t), t=0..2*Pi, a=1..2);
```

laat de functie  $\sin t$  overgaan in  $\sin 2t$ . Er verschijnt een 2D plot-window, met daarin een knoppenbalk. De werking is vergelijkbaar met een video-recorder. De **Stop**- en **Play**-knop spreken voor zich. Met **Adv** gaat de animatie één beeldje verder, **<<** en **>>** bepalen de snelheid van afspelen. De twee rechter knoppen bepalen de richting en de wijze van afspelen, **Once** voor éénmaal, **Loop** voor oneindig.

De 3D animatie, door middel van `animate3d`, werkt op dezelfde wijze.

## 2.7 Functies

In de voorgaande paragrafen zijn we al enkele malen functies tegengekomen met de volgende vorm

```
> f := x^2+2*x-3;
```

Wanneer we nu de functiewaarde willen berekenen in een bepaald punt dan stuiten we op een probleem:

```
> f(1);
```

$$x(1)^2 + 2x(1) - 3$$

Dit kunnen we op twee manieren ondervangen: allereerst door van  $f$  een echte functie te maken met hulp van het commando `unapply`:

```
> f := unapply(f,x);
```

$$f := x \rightarrow x^2 + 2x - 3$$

Deze notatie komt u waarschijnlijk wel bekend voor. We kunnen nu willekeurige functiewaarden berekenen.

Een alternatief is de *arrow operator*, bestaande uit het `-` en `>`-teken, waarmee we functies van één of meerdere variabelen kunnen definiëren:

```
> f := x -> sin(Pi*x);
> u := (x,t) -> sin(x) + t*cos(x);
```

en zelfs

```
> a := n -> int(f(x)*sin(n*Pi*x/l), x=-1..1);
```

waarin  $f$  de hiervoor gedefinieerde functie is.

Op deze functies kunnen alle hiervoor genoemde opdrachten toegepast worden:

```
> diff(f(t),t);  
> int(u(x,t),t=0..1);  
> plot(a(n),n=0..1);
```

Het is aan de gebruiker om uit te maken in hoeverre deze opdrachten zinvol zijn!

### 3 Gewone differentiaalvergelijkingen

Maple kan een groot aantal gewone differentiaalvergelijkingen op analytische wijze oplossen. Veel technieken zoals de methode van de integrerende factor of de Laplace transformatie worden door middel van het commando `dsolve` gebruikt. In sommige gevallen moet u echter zelf een bepaalde methode stapsgewijs toepassen.

In dit hoofdstuk vindt u de commando's die specifiek gebruikt kunnen worden voor gewone differentiaalvergelijkingen. Niet alle commando's zijn standaard Maple. Sommige zijn speciaal voor de practica ontwikkeld, en u zult ze dus niet op andere machines aantreffen.

De commando's worden geïntroduceerd aan de hand van voorbeelden. Wilt u een beschrijving van de exacte syntax, vraag dan de bijbehorende help-file op met het `?` of met behulp van de help-browser.

#### 3.1 `dsolve`: de differential equation solver

Met het commando `dsolve` kan een groot aantal soorten differentiaalvergelijkingen opgelost worden.

##### Eerste orde differentiaalvergelijkingen

lineair	$y' + p(x)y = q(x)$
separabel	$y' = f(x)g(y)$
Bernoulli	$y' + p(x)y = q(x)y^n$
Clairaut	$y = xy' + f(y')$
Riccati	$y' = p(x)y^2 + q(x)y + r(x)$

##### Tweede orde differentiaalvergelijkingen

lineair	$ay'' + by' + cy = d(x)$ , met $a$ , $b$ en $c$ eventueel complex
Euler	$x^2y'' + axy' + by = c(x)$
Bessel	$x^2y'' + (2k + 1)xy' + (\alpha^2x^{2r} + \beta^2)y = 0$

Laten we eerst eens het Maple-equivalent van een differentiaalvergelijking opstellen. We bekijken

$$y'' - 2y' + 2y = 4x^5, \quad y = y(x).$$

Dit wordt in Maple

```
> dv := diff(y(x),x$2) - 2*diff(y(x),x) + 2*y(x) = 4*x^5;
```

Merk op dat expliciet moet worden aangegeven dat  $y$  een functie van  $x$  is. We lossen deze differentiaalvergelijking op met `dsolve`:

```
> dsolve(dv,y(x));
```

Maple geeft de onbepaalde constanten in het antwoord aan met  $_C1$  en  $_C2$ . Wanneer we het beginwaardeprobleem

$$y'' - 2y' + 2y = 4x^5, \quad y(0) = 0, \quad y'(0) = 1, \quad y = y(x).$$

willen oplossen dan moeten de beginvoorwaarden als volgt worden meegegeven:

```
> dsolve({dv,y(0)=0,D(y)(0)=1},y(x));
```

De afgeleide  $y'(0)$  wordt ingevoerd als  $D(y)(0)$ . Algemener: de  $n^{de}$  afgeleide van  $y$  naar  $x$  in 0 is in deze notatie  $(D@@n)(y)(0)$ .

We kunnen het antwoord verifiëren door het in de oorspronkelijke vergelijking in te vullen:

```
> opl := dsolve(dv,y(x));
> subs(opl,dv);
```

```
> simplify(%);
```

Het is aan te raden antwoorden op deze manier te controleren.

Als extra optie kan de wijze waarop Maple de DV moet oplossen aan `dsolve` worden meegegeven. Zo lost

```
> dsolve(dv,y(x),laplace);
```

de DV op met hulp van de Laplace-transformatie. Andere opties zijn:

```
explicit  y(x) expliciet weergeven
series    oplossen met hulp van reeksontwikkeling
numeric   oplossen met hulp van numerieke methoden.
```

## 3.2 Plotten

Er zijn verschillende manieren om de oplossing van een differentiaalvergelijking in beeld te brengen. Allereerst kan de uitkomst van `dsolve` in beeld gebracht worden:

```
> opl := dsolve({dv,y(0)=0,D(y)(0)=1},y(x));
> plot(rhs(opl),x=0..2);
```

Let hierbij op twee dingen: ten eerste mogen er geen symbolische konstanten in `opl` voorkomen, hetgeen meestal inhoudt dat we een beginwaardeprobleem oplossen. Ten tweede gebruiken we alleen de *right hand side* van `opl`.

Wanneer de oplossing uit meerdere uitdrukkingen bestaat dan kunnen deze met `opl[1]`, `opl[2]`, etc., benaderd worden.

Het is mogelijk dat de oplossing niet in expliciete vorm te geven is. Zo wordt de oplossing van

$$2x + y^2 + 2xyy' = 0, y = y(x)$$

gegeven door

$$x^2 + xy^2 = c$$

met  $c$  een willekeurige konstante.

Een dergelijke impliciet gegeven functie kan met behulp van `implicitplot` getekend worden:

```
> implicitplot(x^2+x*y^2=1,x=-2..5,y=-5..5,grid=[50,50]);
```

De optie `grid` bepaalt de resolutie van de grafiek.

We kunnen echter ook de differentiaalvergelijking zelf gebruiken als invoer van een plot commando. De bibliotheek `DEtools` bevat hiervoor een groot aantal procedures, te weten `DEplot`, `DEplot3d`, `dchange`, `PDEplot`, `dfieldplot`, `phaseportrait`.

Deze library wordt tijdens de praktica automatisch bij het opstarten van Maple geladen, elders moet u hiervoor het commando `with(DEtools)`; geven.

Hier beperken we ons tot het commando `DEplot` en . Daarmee kunnen voor de eerste orde differentiaalvergelijking

$$y' = f(x, y), y(x_0) = y_0$$

de volgende grafieken getekend worden:

```
oplossing BWP          DEplot(dv,y(x),x=-1..1,{[y(0)=1]},arrows=NONE);
voorbeeld richtingsveld  dfieldplot(diff(y(x),x)=y(x),y(x),x=-3..3,
y=-3..2,color=x^2+y^2);
oplossing BWP + richtingsveld  DEplot(diff(y(x),x)+y(x)=0,y(x),x=-1..1,{[y(0)=1]});
```

De optie `arrows` kan de waarde `THIN`, `SLIM`, `THICK`, `LINE` of `NONE` gegeven worden.

### 3.3 Picard iteratie

De oplossing van het beginwaardeprobleem

$$y' = f(x, y), \quad y(x_0) = y_0$$

kan met een Picard-iteratieproces benaderd worden, waarvoor door ons een kleine procedure is gemaakt. Als die nog niet is geladen moet dat eerst gebeuren (hier wordt aangenomen dat deze in de huidige werkdirectory staat).

Daarna kan met het commando “picard” de  $n^{\text{de}}$  stap van dit iteratieproces berekend worden.

```
> read('picard.txt'):
> picard(f(x,y), [x,y], [x0,y0], n);
```

Let op: het commando `picard` is dus *geen* standaard Maple-commando !

Het kan soms aardig zijn deze benadering met een  $n^{\text{de}}$  orde reeksontwikkeling van de oplossing te vergelijken:

```
> series(y(x), x=x0, n);
```

De default waarde voor  $n$  is 6. De oplossing  $y(x)$  moet vanzelfsprekend wel bekend zijn.

### 3.4 De Laplace transformatie

Behalve met de optie `laplace` bij het commando `dsolve` kan een differentiaalvergelijking ook opgelost worden door zelf de Laplace transformatie toe te passen. Met

```
> with(inttrans):
```

wordt de bibliotheek met integraal transformaties geladen als dat nog niet gebeurd is.

Met :

```
> F(s):=laplace(f(t), t, s);
```

wordt de Laplace transformatie van  $f(t)$  berekend, en met

```
> f(t):=invlaplace(F(s), s, t);
```

wordt de inverse transformatie uitgevoerd.

Het oplossen van het beginwaardeprobleem

$$y'' + y' - 12y = 0, \quad y(0) = 1, \quad y'(0) = -1.$$

gaat dan als volgt. Definiëer de vergelijking

```
> dv := diff(y(t), t$2) + diff(y(t), t) - 12*y(t) = 0;
```

en bereken de Laplace getransformeerde:

```
> stap1 := laplace(dv, t, s);
```

Substitueer vervolgens  $Y(s)$  en de beginvoorwaarden met `subs`

```
> stap2 := subs({laplace(y(t), t, s)=Y(s), y(0)=1, D(y)(0)=-1}, stap1);
```

en los hieruit  $Y(s)$  op met hulp van `solve`.

```
> stap3 := solve(stap2,Y(s));
```

Pas tot slot de inverse transformatie toe:

```
> opl := invlaplace(stap3,s,t);
```

Het zal blijken dat deze methode soms de voorkeur heeft, vooral wanneer in het rechterlid van de differentiaalvergelijking stap- of puls-functies voorkomen.

Discontinuïteiten kunnen het best met stapfuncties beschreven worden. Maple heeft hiervoor de Heaviside-functie die als volgt gedefinieerd is:

$$\text{Heaviside}(x) = \begin{cases} 0 & \text{voor } x < 0 \\ 1 & \text{voor } x \geq 0 \end{cases}$$

Hiermee kunnen we ook een functie tot een bepaald interval beperken. Een sinus tussen 0 en  $2\pi$ :

```
> f := x -> (Heaviside(x) - Heaviside(x-2*Pi))*sin(x);
```

Een eenheidspuls op tijdstip  $t_0$  wordt gegeven door  $\text{Dirac}(t-t_0)$ .

### 3.5 Stelsels differentiaalvergelijkingen

Stelsels differentiaalvergelijkingen kunnen in principe met behulp van het `dsolve` commando opgelost worden. Dit soort problemen kan beter in matrix-vector vorm herschreven worden,

$$\mathbf{x}' = \mathbf{A} \mathbf{x} + \mathbf{g}(t)$$

en met behulp van eigenwaarden en -vectoren opgelost worden. Dit vergt enige kennis van lineaire algebra.

Het basismateriaal voor deze methode bestaat dus uit matrices en vectoren. In Maple zijn er speciale commando's nodig om dit soort variabelen van andere te onderscheiden, welke kunnen worden geladen (als dat nog niet is gebeurd) met :

```
> with(linalg):
```

Zo is

```
> A := matrix([[1,2,3],[4,5,6]]);
```

de (2x3)-matrix  $A = \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}$ , en

```
> b := vector([7,8,9]);
```

de kolomvector  $b = \begin{bmatrix} 7 \\ 8 \\ 9 \end{bmatrix}$ , welke door Maple verwarrenderwijs met de rijvector  $b = [ 7 \ 8 \ 9 ]$

wordt weergegeven.

Let op: Als het om een of andere reden niet werkt, kan het zijn dat je de library voor lineaire algebra nog moet laden: `> with(linalg):`

Elementen van deze variabelen kunnen verkregen worden met  $A[i, j]$  en  $b[i]$ .

De volgende arithmetische bewerkingen zijn gedefinieerd:



som	<code>evalm(A+B)</code>
verschil	<code>evalm(A-B)</code>
produkt	<code>evalm(A&amp;*B)</code>
scalair produkt	<code>evalm(k*A)</code>
machtverheffen	<code>evalm(A^n)</code> .

Vanzelfsprekend moeten de dimensies van de variabelen kloppen. Het commando `evalm` is noodzakelijk om Maple tot daadwerkelijke evaluatie te brengen.

Bewerkingen op een enkele matrix worden wel direct geëvalueerd:

$A^T$	<code>transpose(A)</code>
$\bar{A}^T$ of $A^*$	<code>htranspose(A)</code>
$A^{-1}$	<code>inverse(A)</code>
$ A $	<code>det(A)</code>

Een belangrijke rol bij het oplossen van stelsels differentiaalvergelijkingen spelen de eigenwaarden en eigenvectoren van de matrix  $\mathbf{A}$ . De eigenwaarden van een matrix  $\mathbf{A}$  zijn in feite de nulpunten van de determinant van de karakteristieke matrix,

```
> solve(det(charmat(A,r)),r);
```

ofwel de nulpunten van het karakteristiek polynoom in de variabele  $r$  van  $\mathbf{A}$

```
> solve(charpoly(A,r),r);
```

Ze zijn ook te berekenen met

```
> eigenvals(A);
```

Het commando

```
> eigenvects(A);
```

geeft u voor alle eigenwaarden respectievelijk de eigenwaarde, de algebraïsche multipliciteit, en de bijbehorende eigenvector.

Het oplossen van een stelsel vergelijkingen  $\mathbf{A} \mathbf{x} = \mathbf{b}$  geschiedt met

```
> linsolve(A,b);
```

Wanneer de coëfficiënten van de matrix  $\mathbf{P}$  een functie van  $t$  zijn dan spreken we van een *matrix functie*. De afgeleide van  $\mathbf{P}$  is dan gedefinieerd als de afgeleide naar  $t$  van ieder element  $p_{ij}$  van  $\mathbf{P}$

$$\frac{d\mathbf{P}}{dt} = \left[ \frac{dp_{ij}}{dt} \right]$$

en de integraal van  $\mathbf{P}$  als de integraal van ieder element  $p_{ij}$

$$\int_a^b \mathbf{P}(t) dt = \left[ \int_a^b p_{ij}(t) dt \right].$$

In Maple betekent dit dat het `diff` of `int` commando op ieder element van  $\mathbf{P}$  moet worden toegepast. Dit gebeurt met behulp van de functie `map`. Bovenstaande bewerkingen worden dan respectievelijk

```
> map(diff,P,t);
```

en

```
> map(int,P,t);
```

Ook operaties als `simplify`, `expand` en `normal` moeten op deze manier worden toegepast.

De fundamentele matrix  $\Phi(t)$ , de oplossing van het beginwaardeprobleem

$$\Phi' = \mathbf{A}\Phi, \Phi(0) = \mathbf{I},$$

ook wel aangeduid met  $\exp(\mathbf{A}t)$ , kan in Maple in één stap berekend worden met `exponential(A,t)`.

## 4 Partiële differentiaalvergelijkingen

In dit hoofdstuk vindt u de commando's die specifiek gebruikt kunnen worden bij partiële differentiaalvergelijkingen. Niet alle commando's zijn standaard Maple. Sommige zijn speciaal voor de practica ontwikkeld, en u zult ze dus niet op andere machines aantreffen.

De commando's worden geïntroduceerd aan de hand van voorbeelden. Wilt u een beschrijving van de exacte syntax, vraag dan de bijbehorende help-file op met het `?`.

### 4.1 Algemene commando's

Tijdens berekeningen kan het voorkomen dat er bepaalde aannames over variabelen gedaan moeten worden om tot een antwoord te komen. Met `assume(n, integer)` kunnen we er bijvoorbeeld voor zorgen dat Maple bij de berekeningen  $n$  als geheel getal beschouwd. Wanneer een variabele een eigenschap heeft dan wordt dit weergegeven door een er achter geplaatste tilde:  $n \sim$ . Welke eigenschap dat precies is geeft `about(n)` weer.

Met `> n := 'n'`; worden de eigenschappen weer verwijderd.

Periodieke functies kunnen gemaakt worden met de door ons gemaakte procedure `rs`, hetgeen staat voor `rescale`. Willen we bijvoorbeeld een puls-functie creëren, die als volgt gedefinieerd is:

$$f(x) = \begin{cases} 0 & \text{voor } -1 < x \leq 0 \\ 1 & \text{voor } 0 < x \leq 1 \end{cases}, f(x) = f(x + 2)$$

dan kan dat (mits de procedure "rs" in Maple is geladen) met

```
> f := x -> Heaviside(rs(x,-1..1));
```

Door het commando `rs` worden alle functiewaarden terugschaald naar het interval  $(-1,1]$ .

### 4.2 Transformatie van variabelen

Het Maple commando `dchange` voert een transformatie uit op een eerste of tweede orde partiële differentiaalvergelijking. Als voorbeeld transformeren we de Laplace vergelijking van Cartesische naar poolcoördinaten:

$$u_{xx} + u_{yy} = 0, x = r \cos \phi, y = r \sin \phi.$$

```
> PDE := diff(u(x,y),x,x) + diff(u(x,y),y,y) = 0;
> trf:={x=r*cos(phi),y=r*sin(phi)};
> PDE_t := dchange(trf,PDE, u(x,y), {r,phi}, simplify);
```

Als parameters moeten achtereenvolgens meegegeven worden: de twee *vergelijkingen* die de transformatie beschrijven, de partiële differentiaalvergelijking, de functie die getransformeerd moet worden en variabelen in de transformatie die slechts als parameters beschouwd moeten worden.

### 4.3 Fourier reeksen

Laat de functie  $f$  in een Fourier reeks te ontwikkelen zijn:

$$f(x) = \frac{a_0}{2} + \sum_{m=1}^{\infty} a_m \cos\left(\frac{m\pi x}{l}\right) + b_m \sin\left(\frac{m\pi x}{l}\right).$$

Om enkele van de coëfficiënten te berekenen kan de door ons gemaakte procedure `fourier_series` gebruikt worden. Met

```
> fourier_series(x,x,x=-1..1,5);
```

worden de eerste vijf termen van de Fourier-ontwikkeling van  $f(x) = x$  op het interval  $[-l, l)$ , inclusief  $a_0$ , berekend. De parameters zijn: de functie  $f$ , de onafhankelijke variabele in  $f$ , het interval waarop de reeks berekend wordt (aangenomen wordt dat  $f$  hierbuiten periodiek is voortgezet), en het aantal termen van de reeks (dit mag geen onbekend aantal zijn).

De algemene coëfficiënten kunnen voor dezelfde functie  $f$  als hierboven berekend worden met de door ons gemaakte procedure `fourier_alg` :

```
> fourier_alg(x,x,x=-1..1);
```

Uitvoer zijn de coëfficiënten  $a_0$ ,  $a_m$  en  $b_m$ .

Het grafische verloop van de Fourier-ontwikkeling kan bekeken worden met de door ons gemaakte procedure `gibbs`. Hiermee wordt zowel  $f$  als zijn Fourier-ontwikkeling in een plot afgebeeld, en dit voor enkele waarden van  $m$ . Voor bovenstaande functie:

```
> read('gibbs.txt');
> gibbs(x,x,x=-1..1,20);
```

Een lijst toont voor welke 8 waarden tussen 0 en 20 de ontwikkeling getekend wordt. Symbolische grenzen voor  $x$  zijn hier *niet* toegestaan.

#### 4.4 Eerste orde partiële differentiaalvergelijkingen

Voor eerste orde quasi-lineaire partiële differentiaalvergelijkingen zijn er verschillende standaard Maple en door ons gemaakte procedures om de oplossing in beeld te brengen.

Allereerst is er `PDEplot`. Dit *tekent* de oplossing, het *berekent* hem dus niet.

`PDEplot` zit wel standaard in Maple, maar is niet direct beschikbaar en zal eerst moeten worden geladen als dat nog niet gebeurd is.

Als voorbeeld nemen we de vergelijking

$$u_x + u_t = -u, \quad u(x, 0) = 1 + \cos x, \quad u(x, t).$$

De aanroep van `PDEplot` wordt dan

```
> with(PDETools,PDEplot):
> pde := diff(u(x,t),x) + diff(u(x,t),t) = - u(x,t);
> F := x -> 1 + cos(x);
> PDEplot(pde, [x,0,F(x)], u(x,t), x=-Pi..Pi, t=0..5);
```

De eerste parameter is de PDE zelf. De tweede parameter is een parametrisatie van de beginkromme. De vierde en vijfde geven het bereik van de karakteristieke coördinaten waarover de oplossing getekend moet worden.

Voor de overige optionele parameters wordt u naar de helppagina voor `PDEplot` verwezen.

Door het oplossen van het karakteristieke stelsel

$$\frac{dx}{ds} = P(x, y, u), \quad \frac{dt}{ds} = Q(x, y, u), \quad \frac{du}{ds} = R(x, y, u)$$

met bijbehorende beginvoorwaarden

$$x(s=0) = f_1(\lambda), \quad t(s=0) = f_2(\lambda), \quad u(s=0) = f_3(\lambda)$$

vinden we een parametrische uitdrukking

$$x = x(s, l), \quad t = t(s, l), \quad u = u(s, l)$$

voor het oplossingsoppervlak. Deze oplossing kan aan verschillende door ons ontwikkelde (dus niet standaard) Maple procedures worden aangeboden, die uiteraard weer vooraf moeten worden geladen indien dat nog niet gebeurd is.

Zo worden met

```
> PDEplotchar([s+1,s],[s,l],s=0..5,l=-2..2,20);
```

twintig grond-karakteristieken voor het vorige voorbeeld getekend, en met

```
> PDEplot3Dchar([s+1,s,exp(-s)*(1+cos(l))],[s,l],s=0..5,l=-2..2,20);
```

twintig ruimtelijke karakteristieken.

PDEanimate3D maakt een animatie van het opspannen van het oplossingsoppervlak met hulp van de ruimtelijke karakteristieken.

```
> PDEanimate3D([s+1,s,exp(-s)*(1+cos(l))],[s,l],s=0..5,l=-2..2,
  steps=[20,20]);
```

De vijfde parameter is het aantal stappen dat in  $s$  en  $l$  richting moet worden genomen. Als extra optie kan de richting waarin het oppervlak gegenereerd wordt gegeven worden.

## Alle gebruikte commando's op een rijtje

De met een \* gemerkte commando's zijn *geen* standaard Maple-commando's.

>	3	grid	10, 14
?	5	Heaviside	15
;	6	htranspose	16
:	6	I	5
"	6	implicitplot	13
{ }	7, 10	infinity	5
->	11	int	8
about	18	Int	8
abs	5	inverse	16
animate	10	invlaplace	14
animate3d	10	laplace	14
arrows	14	linsolve	17
assume	18	map	17
charmat	16	matrix	16
charpoly	16	normal	8
combine	8	PDEanimate3D *	20
D	13	PDEplot	19
dchange	18	PDEplot3Dchar *	20
DEplot	14	PDEplotchar *	20
det	16	Pi	5
diff	9	picard *	14
Dirac	15	plot	10
dsolve	12	plot3d	10
E	5	restart	4
eigenvals	16	rhs	13
eigenvects	16	rs *	18
evalf	6	series	14
evalm	16	simplify	6
exp	5	solve	7
expand	7	sort	8
exponential	17	sqrt	5
factor	8	transpose	16
fourier_alg *	19	subs	8
fourier_series *	19	unapply	11
fsolve	7	vector	16
gibbs *	19		